Pensieve header: The category GDO and the Heisenberg algebra. Based on AlexanderFromHeisenberg.nb at pensieve://Talks/LearningSeminarOnCategorification-2006/.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Classes\\21-1350-KnotTheory"];
<< Common.m
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

A hack to fix how Expand acts on Series:

In[∘]:=
```
Unprotect[SeriesData];
SeriesData /: Expand[sd_SeriesData] := MapAt[Expand, sd, 3];
Protect[SeriesData];
```

In[∘]:=
```
{p*, x*, π*, ξ*} = {π, ξ, p, x}; (u_i_)* := (u*)_i; l_List* := #* & /@ l;
```

In[∘]:=
```
(𝔾_{A1_→B1_}[L1_] // 𝔾_{A2_→B2_}[L2_]) /; (B1* === A2) := Module[{ν, r, x, ξ},
  r = Expand[(L1 /. Table[x → ν[x], {x, B1}]) (L2 /. Table[ξ → ν[ξ], {ξ, A2}])];
  Do[r = r /. ν[x]^p_. ν[x*]^p_. → p! /. ν[x | x*] → 0, {x, B1}];
  𝔾_{A1→B2}[r]
 ]
```

In[∘]:= $\mathbb{G}_{\{\xi_1\}\to\{x_1\}}\left[1+\xi_1 x_1+\xi_1^2 x_1^2\right]$ // $\mathbb{G}_{\{\xi_1\}\to\{x_2\}}\left[1+\xi_1 x_2+\xi_1^2 x_2^2\right]$

Out[∘]= $\mathbb{G}_{\{\xi_1\}\to\{x_2\}}\left[1+x_2\,\xi_1+2\,x_2^2\,\xi_1^2\right]$

In[∘]:=
```
𝔼_{s1_}[ωQ1__] ≡ 𝔼_{s2_}[ωQ2__] := s1 === s2 ∧ Simplify[{ωQ1} == {ωQ2}]
```

In[∘]:=
```
𝔼_{A1_→B1_}[ω1_, Q1_] 𝔼_{A2_→B2_}[ω2_, Q2_] ^:= 𝔼_{A1⋃A2→B1⋃B2}[ω1 ω2, Q1 + Q2]
```

In[∘]:=
```
CF = ExpandNumerator@*ExpandDenominator@*PowerExpand@*Factor;
```

In[∘]:=
```
(𝔼_{A1_→B1_}[ω1_, Q1_] // 𝔼_{A2_→B2_}[ω2_, Q2_]) /; (B1* === A2) :=
 Module[{i, j, E1, F1, G1, E2, F2, G2, I, M = Table},
  I = IdentityMatrix@Length@B1;
  E1 = M[∂_{i,j}Q1, {i, A1}, {j, B1}]; E2 = M[∂_{i,j}Q2, {i, A2}, {j, B2}];
  F1 = M[∂_{i,j}Q1, {i, A1}, {j, A1}]; F2 = M[∂_{i,j}Q2, {i, A2}, {j, A2}];
  G1 = M[∂_{i,j}Q1, {i, B1}, {j, B1}]; G2 = M[∂_{i,j}Q2, {i, B2}, {j, B2}];
  𝔼_{A1→B2}[CF[ω1 ω2 Det[I - F2.G1]^(-1/2)], CF@Plus[
     If[A1 === {} ∨ B2 === {}, 0, A1.E1.Inverse[I - F2.G1].E2.B2],
     If[A1 === {}, 0, 1/2 A1.(F1 + E1.F2.Inverse[I - G1.F2].E1ᵀ).A1],
     If[B2 === {}, 0, 1/2 B2.(G2 + E2ᵀ.G1.Inverse[I - F2.G1].E2).B2]]]]
```

In[∘]:=
```
Random𝔼_{A_→B_}[r_] := Module[{ri},
  ri := RandomInteger[{-r, r}];
  𝔼_{Table[ξ_i,{i,A}]→Table[x_j,{j,B}]}[
   ri,
   Sum[ri ℏ ξ_i ξ_j, {i, A}, {j, A}] +
    Sum[ri ℏ ξ_i x_j, {i, A}, {j, B}] + Sum[ri ℏ x_i x_j, {i, B}, {j, B}]
   ]]
```

In[∘]:= **E1 = Random𝔼**$_{\{1,2\}\to\{1,2,3,4\}}$**[5]**

Out[∘]= $\mathbb{E}_{\{\xi_1,\xi_2\}\to\{x_1,x_2,x_3,x_4\}}\left[-1,\ \hbar x_1^2 - 6\hbar x_1 x_3 + \hbar x_2 x_3 + \hbar x_3^2 + 5\hbar x_1 x_4 - 2\hbar x_2 x_4 - 3\hbar x_3 x_4 - 4\hbar x_4^2 - 4\hbar x_1 \xi_1 - \hbar x_2 \xi_1 - 2\hbar x_3 \xi_1 + 4\hbar x_4 \xi_1 + \hbar \xi_1^2 + 4\hbar x_1 \xi_2 - 3\hbar x_2 \xi_2 + 2\hbar x_3 \xi_2 - 5\hbar x_4 \xi_2 + 3\hbar \xi_1 \xi_2 - 4\hbar \xi_2^2\right]$

In[∘]:= **E2 = Random𝔼**$_{\{1,2,3,4\}\to\{1,2,3\}}$**[5]**

Out[∘]= $\mathbb{E}_{\{\xi_1,\xi_2,\xi_3,\xi_4\}\to\{x_1,x_2,x_3\}}\left[4,\ -5\hbar x_1^2 - \hbar x_1 x_2 + 3\hbar x_2^2 + 2\hbar x_1 x_3 + 6\hbar x_2 x_3 - 4\hbar x_3^2 + 5\hbar x_1 \xi_1 - 2\hbar x_2 \xi_1 + 5\hbar x_3 \xi_1 - 2\hbar \xi_1^2 - 2\hbar x_1 \xi_2 + 5\hbar x_2 \xi_2 - \hbar x_3 \xi_2 + 7\hbar \xi_1 \xi_2 - 4\hbar \xi_2^2 + 5\hbar x_1 \xi_3 - 3\hbar x_2 \xi_3 + 5\hbar x_3 \xi_3 + 9\hbar \xi_1 \xi_3 + 3\hbar \xi_2 \xi_3 + \hbar \xi_3^2 - \hbar x_1 \xi_4 + \hbar x_2 \xi_4 - 5\hbar x_3 \xi_4 + \hbar \xi_1 \xi_4 - 2\hbar \xi_2 \xi_4 - 3\hbar \xi_3 \xi_4 + 4\hbar \xi_4^2\right]$

In[∘]:=
```
E1 // E2
```

Out[∘]=

$$\mathbb{E}_{\{\xi_1,\xi_2\}\to\{x_1,x_2,x_3\}}\Big[-\frac{4}{\sqrt{1+134\,\hbar^2+13\,773\,\hbar^4+290\,412\,\hbar^6+618\,164\,\hbar^8}},$$
$$\big(-5\,\hbar\,x_1^2-798\,\hbar^3\,x_1^2-82\,119\,\hbar^5\,x_1^2-2\,338\,024\,\hbar^7\,x_1^2-3\,549\,114\,\hbar^9\,x_1^2-\hbar\,x_1\,x_2+30\,\hbar^3\,x_1\,x_2+3713\,\hbar^5\,x_1\,x_2+$$
$$684\,034\,\hbar^7\,x_1\,x_2-1\,705\,280\,\hbar^9\,x_1\,x_2+3\,\hbar\,x_2^2+349\,\hbar^3\,x_2^2+35\,391\,\hbar^5\,x_2^2+614\,168\,\hbar^7\,x_2^2+$$
$$3\,009\,060\,\hbar^9\,x_2^2+2\,\hbar\,x_1\,x_3-69\,\hbar^3\,x_1\,x_3-10\,656\,\hbar^5\,x_1\,x_3-1\,122\,643\,\hbar^7\,x_1\,x_3+117\,238\,\hbar^9\,x_1\,x_3+$$
$$6\,\hbar\,x_2\,x_3+1039\,\hbar^3\,x_2\,x_3+111\,245\,\hbar^5\,x_2\,x_3+2\,792\,034\,\hbar^7\,x_2\,x_3+3\,190\,976\,\hbar^9\,x_2\,x_3-4\,\hbar\,x_3^2-$$
$$801\,\hbar^3\,x_3^2-90\,229\,\hbar^5\,x_3^2-2\,250\,955\,\hbar^7\,x_3^2-3\,697\,158\,\hbar^9\,x_3^2-32\,\hbar^2\,x_1\,\xi_1-2564\,\hbar^4\,x_1\,\xi_1-$$
$$114\,700\,\hbar^6\,x_1\,\xi_1+4\,434\,312\,\hbar^8\,x_1\,\xi_1+13\,\hbar^2\,x_2\,\xi_1+693\,\hbar^4\,x_2\,\xi_1-59\,440\,\hbar^6\,x_2\,\xi_1-5\,054\,748\,\hbar^8\,x_2\,\xi_1-$$
$$49\,\hbar^2\,x_3\,\xi_1-4649\,\hbar^4\,x_3\,\xi_1-117\,264\,\hbar^6\,x_3\,\xi_1+3\,801\,100\,\hbar^8\,x_3\,\xi_1+\hbar\,\xi_1^2+288\,\hbar^3\,\xi_1^2+39\,983\,\hbar^5\,\xi_1^2+$$
$$1\,196\,220\,\hbar^7\,\xi_1^2-5\,665\,092\,\hbar^9\,\xi_1^2+41\,\hbar^2\,x_1\,\xi_2+4904\,\hbar^4\,x_1\,\xi_2+298\,997\,\hbar^6\,x_1\,\xi_2+3\,585\,030\,\hbar^8\,x_1\,\xi_2-$$
$$34\,\hbar^2\,x_2\,\xi_2-4300\,\hbar^4\,x_2\,\xi_2-267\,086\,\hbar^6\,x_2\,\xi_2-3\,644\,020\,\hbar^8\,x_2\,\xi_2+58\,\hbar^2\,x_3\,\xi_2+7724\,\hbar^4\,x_3\,\xi_2+$$
$$328\,454\,\hbar^6\,x_3\,\xi_2+3\,245\,476\,\hbar^8\,x_3\,\xi_2+3\,\hbar\,\xi_1\,\xi_2+194\,\hbar^3\,\xi_1\,\xi_2-14\,521\,\hbar^5\,\xi_1\,\xi_2-1\,234\,412\,\hbar^7\,\xi_1\,\xi_2-$$
$$13\,116\,932\,\hbar^9\,\xi_1\,\xi_2-4\,\hbar\,\xi_2^2-550\,\hbar^3\,\xi_2^2-35\,524\,\hbar^5\,\xi_2^2-759\,846\,\hbar^7\,\xi_2^2-4\,784\,420\,\hbar^9\,\xi_2^2\big)\Big/$$
$$\big(1+134\,\hbar^2+13\,773\,\hbar^4+290\,412\,\hbar^6+618\,164\,\hbar^8\big)\Big]$$

In[∘]:=
```
𝔼Series[𝔼s_[ω_, Q_], d_] := 𝔾s[Expand@Series[ω Exp[Q], {ℏ, 0, d}]]
```

In[∘]:=
```
𝔼Series[E1, 1]
```

Out[∘]=

$$\mathbb{G}_{\{\xi_1,\xi_2\}\to\{x_1,x_2,x_3,x_4\}}\Big[$$
$$-1+\big(-x_1^2+6\,x_1\,x_3-x_2\,x_3-x_3^2-5\,x_1\,x_4+2\,x_2\,x_4+3\,x_3\,x_4+4\,x_4^2+4\,x_1\,\xi_1+x_2\,\xi_1+2\,x_3\,\xi_1-$$
$$4\,x_4\,\xi_1-\xi_1^2-4\,x_1\,\xi_2+3\,x_2\,\xi_2-2\,x_3\,\xi_2+5\,x_4\,\xi_2-3\,\xi_1\,\xi_2+4\,\xi_2^2\big)\,\hbar+O[\hbar]^2\Big]$$

In[∘]:=
```
𝔼Series[E2, 1]
```

Out[∘]=

$$\mathbb{G}_{\{\xi_1,\xi_2,\xi_3,\xi_4\}\to\{x_1,x_2,x_3\}}\Big[$$
$$4+\big(-20\,x_1^2-4\,x_1\,x_2+12\,x_2^2+8\,x_1\,x_3+24\,x_2\,x_3-16\,x_3^2+20\,x_1\,\xi_1-8\,x_2\,\xi_1+20\,x_3\,\xi_1-8\,\xi_1^2-$$
$$8\,x_1\,\xi_2+20\,x_2\,\xi_2-4\,x_3\,\xi_2+28\,\xi_1\,\xi_2-16\,\xi_2^2+20\,x_1\,\xi_3-12\,x_2\,\xi_3+20\,x_3\,\xi_3+36\,\xi_1\,\xi_3+$$
$$12\,\xi_2\,\xi_3+4\,\xi_3^2-4\,x_1\,\xi_4+4\,x_2\,\xi_4-20\,x_3\,\xi_4+4\,\xi_1\,\xi_4-8\,\xi_2\,\xi_4-12\,\xi_3\,\xi_4+16\,\xi_4^2\big)\,\hbar+O[\hbar]^2\Big]$$

In[∘]:=
```
𝔼Series[E1, 1] // 𝔼Series[E2, 1]
```

Out[∘]=

$$\mathbb{G}_{\{\xi_1,\xi_2\}\to\{x_1,x_2,x_3\}}\Big[$$
$$-4+\big(20\,x_1^2+4\,x_1\,x_2-12\,x_2^2-8\,x_1\,x_3-24\,x_2\,x_3+16\,x_3^2-4\,\xi_1^2-12\,\xi_1\,\xi_2+16\,\xi_2^2\big)\,\hbar+O[\hbar]^2\Big]$$

In[∘]:=
```
𝔼Series[E1 // E2, 1]
```

Out[∘]=

$$\mathbb{G}_{\{\xi_1,\xi_2\}\to\{x_1,x_2,x_3\}}\Big[$$
$$-4+\big(20\,x_1^2+4\,x_1\,x_2-12\,x_2^2-8\,x_1\,x_3-24\,x_2\,x_3+16\,x_3^2-4\,\xi_1^2-12\,\xi_1\,\xi_2+16\,\xi_2^2\big)\,\hbar+O[\hbar]^2\Big]$$

In[∘]:=
```
𝔼Series[E1 // E2, 5] == (𝔼Series[E1, 5] // 𝔼Series[E2, 5])
```

Out[∘]= True

In[∘]:=
```
A_ \ B_ := Complement[A, B];
(𝔼_{A1_→B1_}[ω1_, Q1_] // 𝔼_{A2_→B2_}[ω2_, Q2_]) /; (B1* =!= A2) :=
 𝔼_{A1⋃(A2\B1*)→B1⋃A2*}[ω1, Q1 + Sum[ξ* ξ, {ξ, A2 \ B1*}]] //
  𝔼_{B1*⋃A2→B2⋃(B1\A2*)}[ω2, Q2 + Sum[z* z, {z, B1 \ A2*}]]
```

A proof of the formula for R is at http://drorbn.net/cat20.

In[∘]:=
```
R_{i_,j_} := 𝔼_{{}→{p_i,x_i,p_j,x_j}}[T^{1/2}, (1 - T) p_j x_j + (T - 1) p_i x_j];
R̄_{i_,j_} := 𝔼_{{}→{p_i,x_i,p_j,x_j}}[T^{-1/2}, (1 - T^{-1}) p_j x_j + (T^{-1} - 1) p_i x_j];
C_{i_} := 𝔼_{{}→{p_i,x_i}}[T^{1/2}, 0]; C̄_{i_} := 𝔼_{{}→{p_i,x_i}}[T^{-1/2}, 0];
```

In[∘]:=
```
η_{i_} := 𝔼_{{}→{p_i,x_i}}[1, 0]
```

In[∘]:=
```
m_{i_,j_→k_} := 𝔼_{{π_i,ξ_i,π_j,ξ_j}→{p_k,x_k}}[1, -ξ_i π_j + (π_i + π_j) p_k + (ξ_i + ξ_j) x_k]
```

In[∘]:=
```
𝔼_{{}→vs_}[ωi_, Q_]_h := Module[{ps, xs, M},
   ps = Cases[vs, p_]; xs = Cases[vs, x_];
   M = Table[ωi, 1 + Length@ps, 1 + Length@xs];
   M⟦2 ;;, 2 ;;⟧ = Table[CF[∂_{i,j}Q], {i, ps}, {j, xs}];
   M⟦2 ;;, 1⟧ = ps; M⟦1, 2 ;;⟧ = xs;
   MatrixForm[M]_h]
```

Reidemeister 3

In[∘]:=
```
(R_{1,2} R_{4,3} R_{5,6} // m_{1,4→1} m_{2,5→2} m_{3,6→3}) ≡ (R_{2,3} R_{1,6} R_{4,5} // m_{1,4→1} m_{2,5→2} m_{3,6→3})
```

Out[∘]=
```
True
```

Reidemeister 2

In[∘]:=
```
{(R̄_{1,2} R_{3,4} // m_{1,3→1} m_{2,4→2}) ≡ η_1 η_2,
 (R_{1,4} R̄_{5,2} C̄_3 // m_{2,4→2} // m_{1,3→1} // m_{1,5→1}) ≡ C̄_1 η_2}
```

Out[∘]=
```
{True, True}
```

Reidemeister 1’s



In[∘]:= $\{(\overline{C}_2 R_{1,3} \mathbin{//} m_{1,2\to1} \mathbin{//} m_{1,3\to1}) \equiv \eta_1,\ (\overline{C}_2 \overline{R}_{3,1} \mathbin{//} m_{1,2\to1} \mathbin{//} m_{1,3\to1}) \equiv \eta_1,\ (C_2 \overline{R}_{1,3} \mathbin{//} m_{1,2\to1} \mathbin{//} m_{1,3\to1}) \equiv \eta_1,\ (C_2 R_{3,1} \mathbin{//} m_{1,2\to1} \mathbin{//} m_{1,3\to1}) \equiv \eta_1\}$

Out[∘]= {True, True, True, True}

The “First Tangle”



In[∘]:= $\text{Factor} /@ (z = R_{1,6} \overline{C}_3 \overline{R}_{7,4} \overline{R}_{5,2} \mathbin{//} m_{1,3\to1} \mathbin{//} m_{1,4\to1} \mathbin{//} m_{1,5\to1} \mathbin{//} m_{1,6\to1} \mathbin{//} m_{2,7\to2})$

Out[∘]= $\mathbb{E}_{\{\}\to\{p_1,p_2,x_1,x_2\}}\left[\frac{T}{-1+2T}, \frac{(-1+T)(p_1-p_2)(T x_1 - x_2)}{-1+2T}\right]$

In[∘]:= z_h

Out[∘]= $\begin{pmatrix} \frac{T}{-1+2T} & x_1 & x_2 \\ p_1 & \frac{-T+T^2}{-1+2T} & \frac{1-T}{-1+2T} \\ p_2 & \frac{T-T^2}{-1+2T} & \frac{-1+T}{-1+2T} \end{pmatrix}_h$

In[∘]:= ZF[Knot[4, 1]]

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]= $\mathbb{E}_{\{\}\to\{p_1,x_1\}}\left[\frac{T}{1-3T+T^2}, 0\right]$

In[∘]:= Alexander[Knot[4, 1]][T]

Out[∘]= $3 - \frac{1}{T} - T$

In[∘]:= Timing@Table[Simplify[Alexander[K][T] ZF[K][[1]]], {K, AllKnots[{3, 9}]}]

Out[∘]= {25.0469, {1, -1, 1, 1, -1, -1, 1, 1, 1, 1, 1, 1, -1, 1, -1, -1, -1, -1, -1, -1, 1, 1, -1, 1, -1, 1, 1, -1, 1, 1, -1, -1, 1, 1, -1, 1, 1, 1, 1, 1, 1, 1, -1, 1, 1, -1, -1, 1, 1, -1, 1, 1, 1, 1, -1, -1, 1, 1, -1, -1, 1, -1, 1, 1, -1, 1, 1, -1, -1, 1, -1, 1, 1, -1, 1, 1, -1, -1, 1, -1, -1, 1, -1, 1}}

In[∘]:= Timing@ZF[GST48]

Out[∘]= $\left\{10.0313, \mathbb{E}_{\{\}\to\{p_1,x_1\}}\left[\frac{T^8}{1-2T+T^2+2T^4-5T^5+2T^6+7T^7-13T^8+7T^9+2T^{10}-5T^{11}+2T^{12}+T^{14}-2T^{15}+T^{16}}, 0\right]\right\}$